code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_53:城塞都市ヴェロエア
;-------------------------------------------------

;-------------------------------------------------
;隊商宿依頼セクション
;-------------------------------------------------
@QUEST_53_EV115_QUEST_LIST

CALL ASK_MULTI_JUDGE("年期の入った紙に記された依頼", QST_GET_SV_EVENT_FLAGF(82) == 1, "やめておく", 1)
SELECTCASE RESULT
	CASE 0
		CALL QUEST_53_PARTLAVOR_1
	CASEELSE
		RETURN
ENDSELECT

;-------------------------------------------------
;短期依頼　義勇兵 ※高等学院でウルスの話を聞いている
;-------------------------------------------------

@QUEST_53_PARTLAVOR_1
#DIM DYNAMIC TARGET_ID
#DIM DYNAMIC MISSION_ID

PRINTL
PRINTFORML 他の依頼と異なり、傷んで年期が入った紙に目を止める
PRINTL
DRAWLINE
SETCOLOR カラー_緑
PRINTL
PRINTFORML 『義勇兵求む』
PRINTL
RESETCOLOR
PRINTFORML 国境を巡邏するヴェロエア軍の間隙を縫って、北方遊牧民『ウルス』の馬賊が辺境集落を襲撃する事案が頻発している
PRINTFORML 機動力の高いウルス馬賊は短時間で略奪・破壊工作を行い、速やかに緩衝地に撤収する長駆侵入作戦を多用しているという
PRINTL
PRINTFORML ある程度の規模の集落は自警団を組織して対策に当たっているようだが、なかなか厳しいのが現実のようだ
PRINTFORML これらの集落に赴き、義勇兵として襲撃者に対処することが求められている
PRINTFORML 労力に対し得られる報酬は割に合うとは言い難いが、この地方での名声を高めることはできるだろう
PRINTL
CALLF TAG_PRINT(@"難易度：@C:{カラー_注意}@★×10～12@/C@", 1)
PRINTL
PRINTFORML 報酬：10000ディナリ＋状況に応じて報酬追加
PRINTFORML 推奨パーティ人数:5人以上　　　推奨平均戦闘値：120～140以上
PRINTL
DRAWLINE
SETCOLOR カラー_選択不可
PRINTFORML 「馬賊の連中は神出鬼没だが……やもすれば、ウルス軍の正規部隊が出張ってきている可能性もある」
PRINTFORMW 「そこいらの十把一絡げの賊と同一視はしないことだ。十分な備えをして臨んでくれ」
RESETCOLOR
PRINTL
PRINTFORMW …この依頼を受けようか？
PRINTL
CALL ASK_YN("クエストに出発する", "他のクエストを探す")
IF RESULT == 0
	PRINTL
	PRINTFORMW 「この依頼を受ける」と宿主に伝えると、ギルドの担当者に引き合わせられた……
	PRINTL
	SETCOLOR カラー_ライトブルー
	PRINTL
	PRINTFORMW 「……ウルスの連中は、ヴェロエア軍の巡邏隊が不在の隙を上手く突いてくる」
	PRINTFORMW 「たとえヴェロエア軍と遭遇戦になっても、御自慢の機動力で逃散することが多いわけだ」
	PRINTL
	PRINTFORMDW 攻撃側の馬賊は長駆偵察に赴き、巡邏隊が留守なら襲撃、駐屯しているなら撤退を選択できる
	PRINTFORMDW 対し、防衛側としては点在する集落を守る、非常に不利な状況ではあるわけだ
	PRINTL
	PRINTFORMW 「まあ、それがかえって我々にとっては狙い目でもあるんだが」
	PRINTL
	PRINTFORMDW ……というと？
	PRINTL
	PRINTFORML 「つまりだ。巡邏隊が不在、あるいは離れた集落に義勇兵を派遣すれば、
	PRINTFORMW 　必然奴らと出くわす可能性も高くなる、ということだな」
	PRINTL
	PRINTFORMDL どうやって、ギルドはヴェロエア軍の行軍計画を把握しているのか……
	PRINTFORMDW ということは、おそらく飯のタネだろうから聞かないでおくことにした
	PRINTL
	;夜だと休憩してから出発
	IF HO_OCLOCK() >= 1800
		PRINTFORMW 「ただ今日はもう遅い。宿の部屋を取っておくから、一晩ゆっくり休んで明日出発してくれ」
		PRINTFORMW 「代金の事は心配しなくていい、依頼主にツケておくさ」
		RESETCOLOR
		PRINTL
		PRINTFORMW 明日に備え、今日はもう休むことにした…
		PRINTL
		;パーティ全員の体力を回復
		FOR LOCAL, 0, QST_PARTYNUM
			TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
			CALL QST_CHANGE_HP(TARGET_ID, 3000)
		NEXT
		WAIT
	ENDIF
	RESETCOLOR
	PRINTL
	PRINTFORMW ……………………
	PRINTFORMW ………………
	PRINTFORMW …………
	PRINTL
	CALL PRINT_TITLE("義勇兵", "イベント")
	PRINTFORMW 乗合馬車で街道を進み、ラクダに乗り換えて数刻…　%QST_LEADER_NAME()%は指定された集落に到着した
	PRINTFORMW 襲撃に備えたものか、防壁を備えた環濠集落となった街の長と挨拶を交わす
	PRINTL
	PRINTFORMW 既に数組のパーティが同様の依頼を受けて待機しているようだ
	PRINTFORMW 約束の期限まで、しばらくここで用心棒として生活することになるだろう
	PRINTL
	PRINTFORMW %QST_LEADER_NAME()%はウルスの騎馬兵たちの戦術について受けたレクチャーを反復することにした……
	PRINTL
	SETCOLOR カラー_ライトブルー
	PRINTFORMW 「ウルスは遊牧騎馬民族。全国民皆騎兵と言ってもいいくらい、男女問わず幼いころから馬に慣れ親しんでるらしい」
	PRINTL
	PRINTFORMW 「その生い立ちが可能にしたのが、馬上から両手を使っての『騎射』だ」
	PRINTFORMW 「ウルスを象徴する、複合弓を主武装にした『弓騎兵』。これが厄介極まりない」
	PRINTL
	PRINTFORMW 「何せ、人間の脚では到底追いつけない騎馬で逃げながら、体を捻って追手に弓を打ってくるからな」
	PRINTFORMW 「広い平原でつかず離れずの距離を保って走り回りつつ、矢を放ち続けて相手の消耗を誘う」
	PRINTFORMW 「いわゆる一撃離脱の『ウルス式射法』……かつて歩兵中心の野戦軍を恐慌に陥れた、奴らのお家芸だよ」
	PRINTL
	PRINTFORMW 「対抗策？　そうだな……堅い防御で凌ぐか、あるいはバリスタみたいな兵器で近づいてくるところを広範囲で吹き飛ばすか」
	PRINTFORMW 「魔法でもなんでも、とにかく広範囲を連続で叩ければ、それが一番効果的だろうさ」
	PRINTL
	PRINTFORMW 「そして、弓騎兵によって混乱した敵陣に、馬鎧と全身鎧で固めた『重騎兵』が突進、壊乱させる」
	PRINTFORMW 「古式ゆかしきランス・チャージの衝撃力は尋常じゃない。弾き飛ばされて、防陣すら崩されるだろう」
	PRINTFORMW 「奴らが突撃のために距離を離したら要注意だ。むろん、こちらの攻撃も届かないからそのつもりでな」
	PRINTL
	PRINTFORMW 「あとは……もし万一正規軍が出張ってきているなら、もう一種の兵科に出会う可能性がある」
	PRINTFORMW 「北辺の異民族から編入された『竜騎兵』。文字通り、飛竜に跨った騎士だ」
	PRINTL
	PRINTFORMW 「竜と戦ったことは……あるか。なら、人間のチームワークで動く竜を想像するといい」
	PRINTFORMW 「地上の騎兵と連携し、機を見て上空からブレスを吹きつけたり、大質量の急降下攻撃を加えてくるんだ」
	PRINTFORMW 「銃火器が発達してなお、ウルスがその武威を保っている理由のひとつだ。相手にはしたくないもんだな」
	RESETCOLOR
	PRINTL
	PRINTFORMW ……………………
	PRINTFORMW ………………
	PRINTFORMW …………
	PRINTL
	;ミッション選択
	SELECTCASE RAND:(100)
		CASE 0 TO 50
			MISSION_ID = 0
		CASE 51 TO 75
			MISSION_ID = 1
		CASEELSE
			MISSION_ID = 2
	ENDSELECT
	CALL FL_STOPBGM()
	CALL FL_PLAYSE("Slash_4.mp3")
	PRINTFORMW 警鐘の音で叩き起こされ、%QST_LEADER_NAME()%たちは配置につく
	PRINTL
	PRINTFORMW 物見やぐらから荒野の果てを見渡せば、その原因がはっきりした
	PRINTL
	SELECTCASE MISSION_ID
		CASE 0
			CALL FL_PLAYBGM("Boss_0.ogg")
			PRINTFORMW 砂塵を上げ、荒野を駆け抜けてくる、数頭の馬
			PRINTFORMW 鬨の声を上げ、馬上の戦士たちが各々の得物をぎらつかせている
			PRINTFORMW ……どうやら、おいでなすったようだ
			PRINTL
			CALLF TAG_PRINT(@"@C:{カラー_注意}@集落に展開した%QST_LEADER_NAME()%に向け、馬賊たちが襲い掛かってきた！@/C@", 1)
		CASE 1
			CALL FL_PLAYBGM("Boss_0.ogg")
			PRINTFORMW 荒野を駆け抜けてくる、数頭の馬……のように見えたそれは、ただの人でも馬でもなかった
			PRINTFORMW 馬の下半身に、人の上体を据え付けたような魔獣……ケンタウロスの群れだ！
			PRINTFORMW 少し話が違うが、魔物ならば滅ぼすほかにない
			PRINTL
			CALLF TAG_PRINT(@"@C:{カラー_注意}@集落に展開した%QST_LEADER_NAME()%に向け、ケンタウロスたちが襲い掛かってきた！@/C@", 1)
		CASEELSE
			CALL FL_PLAYBGM("Boss_1.ogg")
			PRINTFORMW 荒野を駆け抜けてくる騎馬隊　だが、その数は想定よりかなり多い
			PRINTFORMW さらに、高空に目を転じれば……
			PRINTL
			PRINTFORMW 一同、唖然となった
			CALL FL_PLAYSE("Roar_1.mp3")
			PRINTFORMW 何とそこには――飛竜の背に騎乗した、異邦の騎士の姿
			PRINTFORMW ウルスの精兵『竜騎兵』が、その威容を示していた
			PRINTL
			PRINTFORMW ……間違いない
			PRINTFORMW ウルスの正規部隊、その先駆けだ
			PRINTFORMW 数が多い　厳しい連戦になるが、やるしかないようだ！
			PRINTL
			CALLF TAG_PRINT(@"@C:{カラー_注意}@集落に展開した%QST_LEADER_NAME()%に向け、ウルス軍が襲い掛かってきた！@/C@", 1)
	ENDSELECT
	WAIT
	;白兵戦参加メンバーの初期化
	CALL CC_INIT()
	;攻撃側キャラクターの設定(最大6名)
	FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
		CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
	NEXT
	;防御側キャラクターの設定(最大6名)
	SELECTCASE MISSION_ID
		CASE 0
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス重騎兵", "ウルス馬賊槍兵１", -25)
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス重騎兵", "ウルス馬賊槍兵２", -25)
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス馬賊弓兵１", -25)
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス馬賊弓兵２", -25)
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス馬賊弓兵３", -25)
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス馬賊弓兵４", -25)
		CASE 1
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ケンタウロス", "ケンタウロス１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ケンタウロス", "ケンタウロス２")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ケンタウロス", "ケンタウロス３")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_アーチケンタウロス", "アーチケンタウロス１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_アーチケンタウロス", "アーチケンタウロス２")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_アーチケンタウロス", "アーチケンタウロス３")
		CASEELSE
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス重騎兵", "ウルス重騎兵１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス重騎兵", "ウルス重騎兵２")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス弓騎兵１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス弓騎兵２")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス弓騎兵３")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス弓騎兵４")
	ENDSELECT
	;白兵戦イベント呼出
	CALL CLOSE_COMBAT("[BGM_NOT_AUTOSTOP:1]")
	;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
	IF RESULT == 1
		;戦闘参加メンバー全員に経験値を付与
		CALL CC_ADD_COMBAT_EXP(MISSION_ID * 20)
		;戦闘でのHP増減をパーティの体力に反映
		CALL CC_APPLY_CHARA_HP()
		WAIT
		PRINTL
		;ウルス軍との戦いの場合は連戦
		IF MISSION_ID == 2
			PRINTFORMW 第一波が打ち破られたと見るや、上空で様子を見ていた竜騎兵が戦線に舞い降りる
			CALL FL_PLAYSE("Roar_1.mp3")
			PRINTL
			CALLF TAG_PRINT(@"@C:{カラー_注意}@%QST_LEADER_NAME()%は体勢を整え、第二波と対峙した！@/C@", 1)
			WAIT
			;白兵戦参加メンバーの初期化
			CALL CC_INIT()
			;攻撃側キャラクターの設定(最大6名)
			FOR LOCAL, 0, LIST_COUNT(_LIST_QST_PARTY)
				CALL CC_ADDCHARA("ATTACKER", TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL)))
			NEXT
			;防御側キャラクターの設定(最大6名)
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス竜騎兵", "ウルス竜騎兵１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス竜騎兵", "ウルス竜騎兵２")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス重騎兵", "ウルス重騎兵１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス重騎兵", "ウルス重騎兵２")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス弓騎兵１")
			CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ウルス弓騎兵", "ウルス弓騎兵２")
			;白兵戦イベント呼出
			CALL CLOSE_COMBAT("[BGM_NOT_AUTOSTOP:1]")
			;CLOSE_COMBATの戻り値は攻撃側勝利 = 1,防衛側勝利 = 2
			IF RESULT == 1
				;戦闘参加メンバー全員に経験値70を付与
				CALL CC_ADD_COMBAT_EXP(70)
				;戦闘でのHP増減をパーティの体力に反映
				CALL CC_APPLY_CHARA_HP()
			ELSE
				CALL QUEST_53_PL1_DEFEAT(3)
			ENDIF
		ENDIF
		WAIT
		PRINTL
		PRINTFORML 軍馬の嘶き、響き渡る鏑矢の音とともに、敵勢力が撤退を始める
		PRINTFORMW ……なんとか、防衛に成功したようだ
		PRINTL
		CALL FL_STOPBGM()
		CALL FL_PLAYSE("Fanfare_0.ogg")
		PRINTFORMW ――『集落の防衛』に成功した！
		;報酬獲得処理
		;CALL COLOR_PRINTW("報酬として10,000ディナリを得た", カラー_注意)
		;MONEY += 10000
		SELECTCASE MISSION_ID
			CASE 0
				CALL QST_CHANGE_ITEM("MONEY,20000,名誉点,150,")
				CALL QST_RANDOM_EQUIP_LOOT_WEAPON(1)
			CASE 1
				CALL QST_CHANGE_ITEM("MONEY,25000,名誉点,250,")
				CALL QST_RANDOM_EQUIP_LOOT_RUNE(1)
			CASEELSE
				CALL QST_CHANGE_ITEM("MONEY,40000,名誉点,350,")
				CALL QST_RANDOM_EQUIP_LOOT_WEAPON(2)
				CALL QST_RANDOM_EQUIP_LOOT_RELIC(1)
		ENDSELECT
		PRINTL
		PRINTFORMW ……………………
		PRINTFORMW ………………
		PRINTFORMW …………
		PRINTL
		CALL FL_PLAYBGM("Orient_1.ogg")
		PRINTFORMW 繰り返し感謝を述べる集落の長と別れ、帰路を急ぐ
		PRINTL
		SELECTCASE MISSION_ID
			CASE 0
				PRINTFORMW 集落の平穏が保てたのが何よりだが、この状況が繰り返されるようでは心休まる暇もないだろう……
				PRINTFORMW 少し心晴れぬ気持ちを抱えつつも、%QST_LEADER_NAME()%は一路、ヴェロエアを目指した
			CASE 1
				PRINTFORMW 魔獣の群れが襲撃してきたのは予想外だったが、なんとかなってよかった
				PRINTL
				PRINTFORMW あれが、ヴェロエア東方に根を張っているという『山の魔人』とやらの尖兵だろうか
				PRINTFORMW さらに東方を目指すならば、あのような魔獣たちを打ち倒さねばならないかもしれない……
			CASE 2
				PRINTFORMW ウルス軍は名にし負う手強さだったが、これで出鼻を挫くことくらいはできたはずだ
				PRINTFORMW あとはヴェロエア軍の仕事だろう……
				PRINTL
				PRINTFORMW ひたひたと迫る戦の気配を感じながら、%QST_LEADER_NAME()%は一路、ヴェロエアを目指した
		ENDSELECT
		PRINTL
		WAIT
		;クエスト終了時の決まり文句
		TFLAG:クエスト_完了フラグ = 1
		CALL TRAIN_EXIT()
	ELSE
		CALL QUEST_53_PL1_DEFEAT(MISSION_ID)
	ENDIF
ELSE
	CALL QUEST_53_EV115_QUEST_LIST
ENDIF

;義勇兵イベントで敗北
@QUEST_53_PL1_DEFEAT(MISSION_ID)
#DIM DYNAMIC MISSION_ID

CALL CC_APPLY_CHARA_HP()
CALL CC_ADD_COMBAT_EXP(15)

SELECTCASE MISSION_ID
	CASE 0
		CALL COLOR_PRINTW(@"―― %QST_LEADER_NAME()%は対騎兵戦に対応できず痛手を負ってしまった…", カラー_注意)
	CASE 1
		CALL COLOR_PRINTW(@"―― %QST_LEADER_NAME()%は疾駆する魔獣の攻撃を止められない…", カラー_注意)
	CASEELSE
		CALL COLOR_PRINTW(@"―― %QST_LEADER_NAME()%は精強な軍勢の猛襲の前に蹴散らされてしまった…", カラー_注意)
ENDSELECT
WAIT
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
PRINTFORMW 他のパーティや集落の自警団の奮戦もあり、なんとか襲撃者たちを撃退することはできたが……
PRINTFORMW 引き換えに、集落も大きな被害を被ってしまった
PRINTFORMW 集落の人々は力なくうなだれている……
PRINTL
PRINTFORMW 当然報酬を受け取ることはできず、最低限の保証金を受け取るのみに留まった
PRINTFORMW 必要経費を差し引くと完全に赤字だろう……
PRINTFORMW 次はもっと準備をしてから挑んだほうが良さそうだ
PRINTL
CALL PRINT_NOTICE("治療費や滞在費で30,000ディナリを失った……")
MONEY -= 30000
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;@QST_QUEST_53_ENEMY_DATA
; クエスト固有エネミーの読み込み処理
; クエストで共通的に使い回すエネミーの戦闘データは@CC_ENEMY_DATAに記述されているが
; クエスト側で後付で追加することもできる。(@QST_INIT_QUEST_XXなどで明示的に読込み処理を呼び出すこと)
; 特定のクエストでしか使わないエネミーのデータ（ボスなど）はクエスト側に記述するとよいだろう
; パラメータの詳細は @CC_ENEMY_DATA 参照
; ！！！ENEMY_DEF_NAMEはユニークである必要があるためクエスト側でエネミーを追加する場合はENEMY_DEF_NAMEに'QUEST_XX_'を接頭すること！！！
;-------------------------------------------------
@QST_QUEST_53_ENEMY_DATA
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_ウルス弓騎兵"
		,"MAX_HP", 10000
		,"STR", 110
		,"MAG", 0
		,"ATTACK", 110
		,"DEFENCE", 35
		,"ARMOR", 300
		,"AGILITY", 200
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_魔術師
		,"ENEMY_AI", "[SKILL_USE_RATE:80]"
		,"ENEMY_PASSIVE_SKILL", "[CHASE_POISON:3]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_PARTHIAN_SHOT:2]"
		,"EXP", 7
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_ウルス重騎兵"
		,"MAX_HP", 15000
		,"STR", 130
		,"MAG", 0
		,"ATTACK", 130
		,"DEFENCE", 45
		,"ARMOR", 450
		,"AGILITY", 160
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_高脅威
		,"ENEMY_AI", "[SKILL_USE_RATE:60]"
		,"ENEMY_PASSIVE_SKILL", "[CHASE_POISON:3]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_CHARGE_PREP:3]"
		,"EXP", 7
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_ウルス竜騎兵"
		,"MAX_HP", 20000
		,"STR", 180
		,"MAG", 50
		,"ATTACK", 180
		,"DEFENCE", 50
		,"ARMOR", 500
		,"AGILITY", 160
		,"HATE_RATE", 120
		,"TARGETING", WPN_脅威選定_魔術師
		,"ENEMY_AI", "[SKILL_USE_RATE:25]"
		,"REGIST_POISON", 40
		,"REGIST_STUN", 40
		,"REGIST_BURN", 40
		,"REGIST_CHARM", 20
		,"REGIST_PANIC", 70
		,"REGIST_BLIND", 20
		,"REGIST_DEATH", 100
		,"ENEMY_PASSIVE_SKILL", "[ENEMY_SKILL_DRAGONHEART:1][ENEMY_SKILL_DRAGONSCALE:1]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_SKILL_DRAGON_RISING:1][ACT_FIRE_BREATH:3][ACT_REACT:3]"
		,"EXP", 9
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_ケンタウロス"
		,"MAX_HP", 18000
		,"STR", 150
		,"MAG", 20
		,"ATTACK", 150
		,"DEFENCE", 40
		,"ARMOR", 350
		,"AGILITY", 180
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_高脅威
		,"ENEMY_AI", "[SKILL_USE_RATE:40]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_CHARGE_PREP:4]"
		,"EXP", 7
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_アーチケンタウロス"
		,"MAX_HP", 15000
		,"STR", 120
		,"MAG", 20
		,"ATTACK", 120
		,"DEFENCE", 35
		,"ARMOR", 200
		,"AGILITY", 200
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_魔術師
		,"ENEMY_AI", "[SKILL_USE_RATE:40]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_PARTHIAN_SHOT:2]"
		,"EXP", 7
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_油ミミズ"
		,"MAX_HP", 4000
		,"STR", 60
		,"MAG", 30
		,"ATTACK", 60
		,"DEFENCE", 15
		,"ARMOR", 200
		,"AGILITY", 120
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_魔術師
		,"ENEMY_AI", "[SKILL_USE_RATE:50]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_ACT_SAND_DIVE:1][ENEMY_SELF_IGNITON:2]"
		,"DROP_ITEM_LIST", @"MONEY,500,1000,{ITEM_魔獣の肉片},1,450,"
		,"EXP", 3
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_タールゴーレム"
		,"MAX_HP", 8000
		,"STR", 100
		,"MAG", 50
		,"ATTACK", 100
		,"DEFENCE", 30
		,"ARMOR", 350
		,"AGILITY", 80
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_無作為
		,"ENEMY_AI", "[SKILL_USE_RATE:50]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_ACT_SAND_DIVE:3][ENEMY_SELF_IGNITON:3]"
		,"DROP_ITEM_LIST", @"MONEY,1000,1000,{ITEM_石油},1-2,500,"
		,"EXP", 4
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_オイルウィスプ"
		,"MAX_HP", 4000
		,"STR", 60
		,"MAG", 85
		,"ATTACK", 60
		,"DEFENCE", 15
		,"ARMOR", 200
		,"AGILITY", 180
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_魔術師
		,"ENEMY_AI", "[SKILL_USE_RATE:50]"
		,"REGIST_BURN", 100
		,"ENEMY_ACTIVE_SKILL", "[ACT_FIRE_BREATH:1]"
		,"DROP_ITEM_LIST", @"MONEY,500,1000,{ITEM_マナ結晶},1-2,500,"
		,"EXP", 3
}
{
	DT_ROW_ADD "ENEMY_DATA"
		,"ENEMY_DEF_NAME", "QUEST_53_ペトロスコーピオン"
		,"MAX_HP", 6000
		,"STR", 80
		,"MAG", 40
		,"ATTACK", 80
		,"DEFENCE", 25
		,"ARMOR", 300
		,"AGILITY", 150
		,"HATE_RATE", 100
		,"TARGETING", WPN_脅威選定_低脅威
		,"ENEMY_AI", "[SKILL_USE_RATE:50]"
		,"ENEMY_PASSIVE_SKILL", "[ATK_ADD_POISON:3][CHASE_POISON:4]"
		,"ENEMY_ACTIVE_SKILL", "[ENEMY_ACT_SAND_DIVE:2][ENEMY_SELF_IGNITON:2]"
		,"DROP_ITEM_LIST", @"MONEY,700,1000,{ITEM_カニの肉},1,450,"
		,"EXP", 4
}

repo_name: pervysquirrelz/erafl

path: ERB/QUEST/クエストデータ/QUEST_53_城塞都市ヴェロエア/QUEST_53_城塞都市ヴェロエア_EV115～依頼.ERB

license: unknown


code:
;-------------------------------------------------
;MAPID_16 EV160～
;郊外の窪地
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_53_EVENT_160
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "窪地の庵"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
;最初は非活性、イベント83を見ており、かつイベント160のイベントフラグが1か2の場合に開放される
RESULTS:3 '= "[EVENT_STATE:DEACTIVE][EVENT_ICON:TALK]"

@QST_QUEST_53_EVENT_160
#DIM DYNAMIC EVENT_FLAG

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

PRINTL
IF HO_OCLOCK() < 1800
	PRINTFORMW 荒野の只中の窪地にぽつねんと佇む、一人の男の姿がある……
	IF EVENT_FLAG < 2
		;初回
		IF EVENT_FLAG == 0
			PRINTL
			SETCOLOR カラー_パ緑
			PRINTFORMW 「んー？　なんだね、きみは」
			PRINTL
			PRINTFORMDW 長いスカーフを纏い、この土地特有の装束を纏った、長い髭を蓄えた痩せぎすの男だ
			PRINTFORMDW 長い眉の下から覗く爛々とした視線は、こちらを値踏みするような色を隠しもしない
			PRINTL
			PRINTFORMW 「『石油』の話を聞きに来た？　ほほう、なかなか先見の明があると見える」
			PRINTFORMW 「私の名前はナジブという。元は学院で教鞭を執っていたものだ。折角だ、案内しよう……ついてくるといい」
			PRINTL
			PRINTFORMDW ナジブと名乗った男に導かれ、足を取られながらもたどり着いた窪地の中央
			PRINTFORMDW 黒い泥沼……水たまりから、ふつふつと泡が湧き出ているのが見える
			PRINTFORMDW 柄杓で少し汲み取ると、水よりやや粘性の強い様子が見て取れた
			PRINTL
			PRINTFORMW 「それが、石油さ。今は建築資材や、燃料として細々と使われている程度のものだ」
			PRINTFORMW 「だが……これからの時代に必要なのは固形燃料ではない。私はそいつが、石炭など目ではない可能性を秘めていると確信している」
			PRINTL
			PRINTFORMW 「長期にわたった調査で、ここには大規模な石油層があると見てる。既に土地の権利は確保済みだ」
			PRINTFORMW 「石油を取り出す装置については、既に設計と試作も済んでいる」
			PRINTFORMW 「あとは、装置を組み立てて、人を雇って穴を掘って採掘ができるようにすればいい、んだが……」
			PRINTL
			PRINTFORMDW ナジブは頭をバリバリと掻いて、はあ、と大きなため息を吐いた
			PRINTL
			PRINTFORMW 「どう話をしても、なかなか投資をしてくれる人間が現れなくてね」
			PRINTFORMW 「勢い余って学院を退いてきたはいいものの、今更あちらには頼れないし」
			PRINTL
			PRINTFORMDW ……何という計画性のなさよ。まあ、学者とはそういうものなのかもしれないが
			PRINTL
			PRINTFORMW 「ともあれ、支援者を確保しないことにはお手上げ、という話でね」
			PRINTL
			PRINTFORMDW なんとも胡乱な話だが、一口乗ってみるのもいいだろうか……？
			PRINTL
			RESETCOLOR
		;二回目以降、まだ身バレしてない
		ELSE
			PRINTL
			SETCOLOR カラー_パ緑
			PRINTFORMW 「むう？　きみ、また来たのか」
			PRINTFORMW 「私もあまりヒマではないんだがね」
			PRINTL
			RESETCOLOR
			PRINTFORMW ナジブはため息まじりにこちらに応対してくる
			PRINTL
			PRINTFORMW 石油の話、一口乗ってみるのもいいだろうか……？
			PRINTL
		ENDIF
		CALL ASK_MULTI_JUDGE("身の上を明かす", 1, "やめておく",1)
		SELECTCASE RESULT
			CASE 0
				PRINTL
				PRINTFORMW 自分は西方の地方領主だ、と身の上を明かすと、ナジブの目の色が変わった
				PRINTL
				SETCOLOR カラー_パ緑
				PRINTFORMW 「なんと。人は見かけによらないものだ」
				PRINTFORMW 「きみ……私に投資をしてみないか？　石油の可能性は先ほど話したとおりだ」
				PRINTFORMW 「石油が採掘できれば、莫大な富をもたらすことは約束されたようなものだ」
				PRINTL
				PRINTFORMW 「……まあ、もちろん掘っても何も出てこない、ということもあり得るかもしれない」
				PRINTFORMW 「確約はできないから、今まで断り続けられてきたのも事実だしな」
				PRINTL
				PRINTFORMW 「私に投資をして、もし石油が採掘できるようになったら、その石油を現物で供給しよう」
				PRINTFORMW 「取り分は、産油量の５割……まあ、運べるだけ、と言ったところになるのかもだが」
				PRINTL
				PRINTFORMW 「本拠地から離れた土地の管理は難しいだろうから、この土地の所有権は私が持つことにする」
				PRINTFORMW 「きみに求めるのは初期投資、油井を掘る費用と汲み出す装置の建造費用だ」
				PRINTL
				PRINTFORMW 「もちろん、維持管理費用やらも発生するだろうが、それはまたおいおい、の話だ」
				PRINTFORMW 「……どうだろう。投資のほう、考えてもらえるだろうか？」
				RESETCOLOR
			CASEELSE
				PRINTL
				SETCOLOR  カラー_パ緑
				PRINTFORMW 「では、私は追加の調査に戻るよ。それじゃあ」
				RESETCOLOR
				PRINTL
				PRINTFORMW
				CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
				RETURN
		ENDSELECT
	;身バレ以降、投資の確認
	ELSE
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「ああ、きみか……どうだい、投資の話、考えてもらえただろうか？」
		RESETCOLOR
		PRINTL
	ENDIF
	PRINTFORMW
	CALL ASK_MULTI_JUDGE("投資する", 1, "やめておく",1)
	SELECTCASE RESULT
		CASE 0
			PRINTL
			SETCOLOR カラー_パ緑
			PRINTFORMW 「……そうか！　いや、西方人にも話が分かる人間がいるものだな！」
			PRINTL
			PRINTFORMDW ナジブは興奮気味にこちらの手を握りしめ、ぶんぶんと乱暴な握手をしてくる
			PRINTL
			PRINTFORMW 「これから共同で事業を営む以上、私たちは『友達』だ。期待を裏切りはしないと約束しよう」
			PRINTFORMW 「友よ、これからよろしく頼むぞ！」
			RESETCOLOR
			PRINTL
			CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 3)
			CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
			CALL QST_ACTIVATE_EVENT(161)
		CASE 1
			PRINTL
			SETCOLOR カラー_パ緑
			PRINTFORMW 「……そうか、残念だ」
			PRINTFORMW 「支援者が見つかるまで、私はしばらくここにいる」
			PRINTFORMW 「気が変わったら、また訪ねてみてくれよ」
			RESETCOLOR
			PRINTL
			CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 2)
			RETURN
	ENDSELECT
ELSE
	PRINTFORMW 誰もいない……
	PRINTFORMW また出直すとしよう
	RETURN
ENDIF

;-------------------------------------------------
;MAPID_16 EV161
;郊外の採掘予定地
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_53_EVENT_161
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "採掘予定地"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
;最初は非活性、イベント160のイベントフラグが3の場合に開放される
RESULTS:3 '= "[EVENT_STATE:DEACTIVE][EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_53_EVENT_161
#DIM DYNAMIC EVENT_FLAG
#DIM DYNAMIC INVESTMENT

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

PRINTL
IF HO_OCLOCK() < 1800
	IF EVENT_FLAG == 0
		PRINTFORMW 図面を広げながら、ナジブが話を始めた
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「では、友よ。具体的に投資の話に移ろう」
		PRINTFORMW 「肝心の油井を掘るのに必要な費用は……西方のディナリ換算だと、ざっとこれくらいだな」
		PRINTL
		SELECTCASE MONEY
			CASE IS < 2500000
				PRINTFORMDW ……なかなかとんでもない金額を提示された
			CASE 2500000 TO 3000000
				PRINTFORMDW 払えなくはない……が、身代を傾けかねない額面だ
			CASE IS > 3000000
				PRINTFORMDW ううむ……なかなか大きな出費になりそうだ
		ENDSELECT
		PRINTL
		PRINTFORMW 「まあ、すぐに全額が必要というわけでもないさ」
		PRINTFORMW 「人足の工面や、ここの採掘拠点化、警備員の雇い入れ……等々、少しずつ進めていく必要があるからな」
		PRINTL
		PRINTFORMW 「そうだな……最低500,000ディナリから投資金を拠出していってくれれば、その段階に応じて運用を始めよう」
		PRINTFORMW 「もちろん、今持ち合わせがないなら、また次の機会でも構わないが」
		RESETCOLOR
	ELSE
		PRINTL
		CALL QST_GET_SV_EVENT_FLAG(161)
		SELECTCASE RESULT
			CASE 0 TO 500000
				PRINTFORMW 窪地に採掘所を設けるための資材が集められていっているようだ
				PRINTFORMW 作業員や輸送員が出入りしている様子が見受けられる……
			CASE 500001 TO 1000000
				PRINTFORMW 窪地には採掘場を設けるための資材が集積されている
				PRINTFORMW いくつかの建屋が建ち、穴を掘るための滑車を設えた足場が組み上げられ始めている
				PRINTFORMW 作業員や輸送員が数多く出入りしている様子が見受けられる……
			CASE 1000001 TO 1500000
				PRINTFORMW 窪地には多くの建屋が建ち、煮炊きの煙も見えている……
				PRINTFORMW 穴を掘るための装置が稼働を始め、人が穴の中に吊り下げられていく様子が見えた
				PRINTFORMW 作業員や輸送員がひっきりなしに出入りしている様子が見受けられる……
			CASE 1500001 TO 2000000
				PRINTFORMW 窪地には多くの建屋が建ち、煮炊きの煙も見えている……
				PRINTFORMW 穴を掘るための装置が常に稼働を続け、人が穴の中に吊り下げられていく様子が見えた
				PRINTFORMW 作業員や輸送員、周辺を巡邏する警備員がひっきりなしに出入りしている様子が見受けられる……
			CASEELSE
				PRINTFORMW 窪地には多くの建屋が建ち並び、煮炊きの煙も見えている……
				PRINTFORMW 穴を掘るための装置が採油装置に造り変えられていく様子が見て取れた
				PRINTFORMW 作業員や輸送員、周辺を巡邏する警備員がひっきりなしに出入りしている様子が見受けられる……
		ENDSELECT
		PRINTL
		PRINTFORMW 周囲の喧噪をものともせず、ナジブがこちらに駆け寄ってくる
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「おお、友よ！」
		SIF EVENT_FLAG > 0
			PRINTFORMW 「前より少しだけ賑やかになっただろう？　友の投資のおかげだ」
		PRINTFORMW 「設備の案内はおいおいするが……さて、今回はどうするね？」
		RESETCOLOR
	ENDIF
	PRINTL
	PRINTFORMW 今回の投資額はいくらにしようか……　（所持金:%TOSTR(MONEY, "###,###,##0")%ディナリ　投資額：%TOSTR(EVENT_FLAG, "###,###,##0")%ディナリ／2,500,000ディナリ）
	CALL ASK_MULTI_JUDGE("500,000ディナリ", MONEY > 500000, "1,000,000ディナリ", MONEY > 1000000 && EVENT_FLAG < 1500001, "残り投資額全額", MONEY > (2500000 - EVENT_FLAG), "やめておく", 1)
	SELECTCASE RESULT
		CASE 0
			INVESTMENT = 500000
		CASE 1
			INVESTMENT = 1000000
		CASE 2
			INVESTMENT = 2500000 - EVENT_FLAG
		CASEELSE
			PRINTL
			SETCOLOR カラー_パ緑
			PRINTFORMW 「そうか。次の機会にまた頼むぞ、友よ」
			RESETCOLOR
			RETURN
	ENDSELECT
	PRINTL
	SETCOLOR カラー_パ緑
	IF INVESTMENT >= 2000000
		PRINTFORMW 「なかなか豪気じゃないか、気に入ったぞ！」
	ENDIF
	IF EVENT_FLAG + INVESTMENT >= 2500000
		PRINTFORMW 「%TOSTR(INVESTMENT, "###,###,##0")%ディナリで目標額達成だ……よし、これならいけるぞ！」
		PRINTFORMW 「すぐに作業にかからせよう！」
	ELSE
		PRINTFORMW 「%TOSTR(INVESTMENT, "###,###,##0")%ディナリだな。よし、それではこれを元手に作業を続けるとしよう！」
	ENDIF
	RESETCOLOR
	PRINTL
	PRINTFORMW ナジブは矢継ぎ早に作業員に指示を飛ばしていく……
	MONEY -= INVESTMENT
	CALLF TAG_PRINT(@"@C:{カラー_注意}@%TOSTR(ABS(INVESTMENT),"###,###,##0")%ディナリを失った@/C@", 1)
	CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, EVENT_FLAG + INVESTMENT)
	CALL FL_PLAYSE("Money_0.mp3")
	IF EVENT_FLAG + INVESTMENT >= 2500000
		CALL PRINT_NOTICE(@"投資額：%TOSTR(EVENT_FLAG + INVESTMENT, "###,###,##0")%ディナリ／2,500,000ディナリ　目標額達成！", カラー_黄)
	ELSE
		CALL PRINT_NOTICE(@"投資額：%TOSTR(EVENT_FLAG + INVESTMENT, "###,###,##0")%ディナリ／2,500,000ディナリ", カラー_緑)
	ENDIF
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSE
	PRINTFORMW 誰もいない……
	PRINTFORMW また出直すとしよう
	RETURN
ENDIF

;-------------------------------------------------
;MAPID_16 EV162
;郊外の油田
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_53_EVENT_162
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "ナジブ油田"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
;最初は非活性、イベント161のイベントフラグが2500000の場合に開放される
RESULTS:3 '= "[EVENT_STATE:DEACTIVE][EVENT_ICON:CHESS-ROOK]"

@QST_QUEST_53_EVENT_162
#DIM DYNAMIC EVENT_FLAG

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

;イベント追加のための補足処理
SIF EVENT_FLAG == 1 && HO_ITEM_STOCK:ITEM_石油 > 0
	EVENT_FLAG = 2

PRINTL
IF HO_OCLOCK() < 1800
	PRINTFORMW ひときわ強い石油の匂いが立ち込めている……
	PRINTL
	PRINTFORMW ドワーフ、獣人、様々な人々が出入りするようになった窪地に、%ANAME(MASTER)%は足を踏み入れる
	PRINTFORMW 多くの建屋が立ち並び、小さな集落同然の規模になった窪地には『ナジブ油田』の標識が立てられていた
	PRINTFORMW 施設の中心では、高く組まれた櫓から直下の採掘坑へ滑車を通した桶が吊るされ、石油を汲み出している……
	IF EVENT_FLAG == 0
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「待っていたぞ、友よ」
		PRINTL
		PRINTFORMDW ナジブが数人の同志を伴い、居住まいを正して、こちらを出迎えてくれた
		PRINTL
		PRINTFORMW 「友の援助で、ようやく一定量の採油が見込める所まで油井を掘り進めることができた」
		PRINTFORMW 「全て友のおかけだ。本当に感謝しているよ」
		PRINTL
		PRINTFORMDW こちらは資金を拠出しただけで、本当に頑張ったのはナジブたちだろう
		PRINTL
		PRINTFORMW 「いや。そもそもの始まりは、%ANAME(MASTER)%が私の夢物語に理解を示してくれたからだ」
		PRINTFORMW 「私は善き友に恵まれた。本当に、そう思うよ」
		PRINTL
		PRINTFORMDW 相好を崩したナジブが、改めて、と握手を求めてきた
		PRINTFORMDW %ANAME(MASTER)%はその手をしっかりと握りしめ、異邦の地の友の偉業を最大限に称え――
		PRINTFORMDW 周囲の作業員が、一斉に喝采を上げた
		PRINTL
		PRINTFORMW 「さて……一席設けたいくらいの気持ちだが、友も忙しい身だろう」
		PRINTFORMW 「約束通り、木製の樽に詰めた石油を用意しておいた。輸送員も手配してある、ぜひ持って行ってくれ」
		PRINTL
		PRINTFORMW 「次に来る時までには、また用意しておく。ぜひ訪ねてきてくれよ！」
		PRINTL
		RESETCOLOR
		CALL FL_PLAYSE("Fanfare_0.ogg")
		CALL QST_CHANGE_ITEM("名誉点,500,")
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
	ELSEIF EVENT_FLAG == 1
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「おお、友よ、久しいな！」
		PRINTL
		PRINTFORMDW ナジブがこちらを歓待してくれた……
		PRINTL
		PRINTFORMW 「石油の利用法についても、学院の面々が研究を始めてくれた」
		PRINTFORMW 「より人々の生活が豊かになればいいと思っているよ」
		PRINTL
		RESETCOLOR
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 2)
	ELSEIF EVENT_FLAG == 2
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「おお、友よ、久しいな！」
		PRINTL
		PRINTFORMDW ナジブがこちらを歓待してくれた……
		PRINTL
		PRINTFORMW 「石油の利用法についても、学院の面々が研究を始めてくれた」
		PRINTFORMW 「より人々の生活が豊かになればいいと思っているよ」
		PRINTL
		PRINTFORMW 「ときに、友よ。我らが油田の運営に、今少々問題が持ち上がっていてな」
		PRINTFORMW 「この匂いに引き寄せられたのか、大小の魔獣が周囲をうろつくようになってしまったんだ」
		PRINTL
		PRINTFORMW 「やもすると、石油に含有されるマナを嗅ぎつけたものかもしれないが……」
		PRINTFORMW 「このままでは、採油業務にも影響が出かねない」
		PRINTL
		PRINTFORMW 「そこで、ジンを討伐せしめた友の力を借りたい」
		PRINTFORMW 「具体的には、油田周辺の魔物の駆除を頼みたいのだ」
		PRINTL
		PRINTFORMW 「特殊な性質を持った魔物も多いが、共通して火器の扱いには気を付けて欲しい」
		PRINTFORMW 「燃焼などさせると、周囲に延焼したりして大変なことになるからな」
		PRINTL
		PRINTFORMW 「無論、駆除が成功した際には、僅かばかりだが報酬を支払おう」
		PRINTFORMW 「駆除の状況に応じては、さらに石油の産出量を増やすこともできるかもしれん」
		PRINTFORMW 「油田の今後のためにも、協力願えると助かるよ」
		PRINTL
		RESETCOLOR
		CALL QST_ACTIVATE_EVENT(163)
		CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 3)
	ELSE
		PRINTL
		SETCOLOR カラー_パ緑
		PRINTFORMW 「おお、友よ、久しいな！」
		PRINTL
		PRINTFORMDW ナジブがこちらを歓待してくれた……
		PRINTL
		PRINTFORMW 「石油の利用法についても、学院の面々が研究を始めてくれた」
		PRINTFORMW 「より人々の生活が豊かになればいいと思っているよ」
		PRINTL
		PRINTFORMW 「石油に集る魔物の駆除のほうも考えてもらえるとありがたい」
		PRINTFORMW 「無論、延焼には気をつけてな。よろしく頼むぞ、友よ！」
		PRINTL
		RESETCOLOR
	ENDIF
	CALL QST_CHANGE_ITEM(@"{ITEM_石油},{QUEST_53_OIL_PRODUCTION()},")
	CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSE
	PRINTFORMW 「今日の営業は終了しました」
	PRINTL
	PRINTFORMW 警備兵がうやうやしくそう伝えてきた
	PRINTFORMW また日を改めて出直すとしよう……
	RETURN
ENDIF

;-------------------------------------------------
;QUEST53用 石油産出量出力関数
;あとあと依頼でさらに拡張予定
;-------------------------------------------------
@QUEST_53_OIL_PRODUCTION()
#FUNCTION
#DIM DYNAMIC PRODUCTION

;基本産出量は3、害獣駆除回数/5を加算
IF QST_GET_SV_EVENT_FLAGF(163) < 5
	PRODUCTION = 3
ELSE
	PRODUCTION = 3 + (QST_GET_SV_EVENT_FLAGF(163) / 5)
ENDIF
RETURNF PRODUCTION

;-------------------------------------------------
;MAPID_16 EV163
;油田の害獣駆除
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_53_EVENT_163
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "油田の害獣駆除"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
;最初は非活性、イベント162のイベントフラグが3以上の場合に開放される
RESULTS:3 '= "[EVENT_STATE:DEACTIVE][EVENT_ICON:BATTLE]"

@QST_QUEST_53_EVENT_163
#DIM DYNAMIC EVENT_FLAG
#DIM DYNAMIC TOTAL_POINT
#DIMS PRESET_NAME

;当日の駆除回数
CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT
;全体の駆除ポイントのログ
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
TOTAL_POINT = RESULT

PRINTL
PRINTFORMW ■魔物駆除ポイント：%TOSTR(TOTAL_POINT, "###,###,##0")%pt　　■駆除による石油増産ボーナス　＋{TOTAL_POINT / 5}
PRINTL
IF EVENT_FLAG > 2
	PRINTFORMW 今日の魔物駆除はここまでにしておこう……（3/3）
	RETURN
ELSE
	PRINTFORML …油田周辺の魔物駆除に出かけますか？（{EVENT_FLAG}/3）
	PRINTFORMW （一時間ほどかかり、敗北時にはクエストが終了します）
	PRINTL
	CALL ASK_YN("駆除に出発する", "やめておく")
	IF RESULT == 0
		PRINTL
		CALL PRINT_TITLE("油田の魔物駆除", "イベント")
		;出現する魔物をセレクト
		PRESET_NAME = %S_R("油ミミズ/タールゴーレム/オイルウィスプ/ペトロスコーピオン")%
		;プリセットに応じたテキストを流す
		CALL QST_QUEST_53_EVENT163_MESSAGE_BEFORE_BATTLE(PRESET_NAME)
		PRINTL
		;白兵戦参加メンバーの初期化
		CALL CC_INIT()
		;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
		CALL CC_ADDPARTY()
		CALL QST_QUEST_53_COMBAT_MOB_PRESET(PRESET_NAME)
		CALL CLOSE_COMBAT("[BGM:Battle_0.ogg]")
		PRINTL
		IF RESULT == 1
			CALL CC_APPLY_CHARA_HP()
			CALL PRINT_NOTICE(@"%QST_LEADER_NAME()%は戦闘に勝利した！")
			;戦利品獲得処理
			CALL CC_GET_LOOT_FROM("DEFENDER")
			CALL QST_CHANGE_ITEM("名誉点,5,")
			WAIT
			PRINTL
			CALL FL_PLAYSE("Fanfare_0.ogg")
			CALL COLOR_PRINTW(@"―― %QST_LEADER_NAME()%は油田の周りの魔獣を駆除した…({EVENT_FLAG + 1}/3)", カラー_注意)
			PRINTFORML 　　　■魔物駆除ポイント：%TOSTR(TOTAL_POINT, "###,###,##0")%pt　→　%TOSTR(TOTAL_POINT + 1, "###,###,##0")%pt
			IF (TOTAL_POINT + 1) % 5 == 0
				PRINTFORMW 　　　駆除によって石油が増産されるようになった（供給量：{QUEST_53_OIL_PRODUCTION()} → {QUEST_53_OIL_PRODUCTION() + 1} ）
			ELSE
				WAIT
			ENDIF
			;当日の実行回数に加算
			CALL QST_SET_EVENT_FLAG(QST_NOWEV, EVENT_FLAG + 1)
			;全体のポイントに加算
			CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, TOTAL_POINT + 1)
			HO_TIME += 60
		ELSE
			CALL CC_APPLY_CHARA_HP()
			CALL CC_ADD_COMBAT_EXP(10)
			CALL PRINT_NOTICE(@"%QST_LEADER_NAME()%は戦闘に敗北してしまった……")
			PRINTL
			PRINTFORMW ……………………
			PRINTFORMW ………………
			PRINTFORMW …………
			PRINTL
			CALL PRINT_NOTICE("治療費や滞在費で10,000ディナリを失った……")
			MONEY -= 10000
			TFLAG:クエスト_完了フラグ = 1
			CALL TRAIN_EXIT()
		ENDIF
	ELSE
		RETURN
	ENDIF
ENDIF

;-------------------------------------------------
;プリセットに応じた戦闘前メッセージ
;-------------------------------------------------
@QST_QUEST_53_COMBAT_MOB_PRESET(PRESET_NAME)
#DIMS PRESET_NAME
SELECTCASE PRESET_NAME
	CASE "油ミミズ"
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_油ミミズ", "油ミミズ１")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_油ミミズ", "油ミミズ２")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_油ミミズ", "油ミミズ３")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_油ミミズ", "油ミミズ４")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_油ミミズ", "油ミミズ５")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_油ミミズ", "油ミミズ６")
	CASE "タールゴーレム"
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_タールゴーレム", "タールゴーレム１")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_タールゴーレム", "タールゴーレム２")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_タールゴーレム", "タールゴーレム３")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_タールゴーレム", "タールゴーレム４")
	CASE "オイルウィスプ"
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_オイルウィスプ", "オイルウィスプ１")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_オイルウィスプ", "オイルウィスプ２")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_オイルウィスプ", "オイルウィスプ３")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_オイルウィスプ", "オイルウィスプ４")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_オイルウィスプ", "オイルウィスプ５")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_オイルウィスプ", "オイルウィスプ６")
	CASE "ペトロスコーピオン"
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ペトロスコーピオン", "ペトロスコーピオン１")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ペトロスコーピオン", "ペトロスコーピオン２")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ペトロスコーピオン", "ペトロスコーピオン３")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ペトロスコーピオン", "ペトロスコーピオン４")
		CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_53_ペトロスコーピオン", "ペトロスコーピオン５")
ENDSELECT

;-------------------------------------------------
;プリセットに応じた戦闘前メッセージ
;-------------------------------------------------
@QST_QUEST_53_EVENT163_MESSAGE_BEFORE_BATTLE(PRESET_NAME)
#DIMS PRESET_NAME
SELECTCASE PRESET_NAME
	CASE "油ミミズ"
		PRINTFORMW %QST_LEADER_NAME()%が油田の周りを巡回していると……
		PRINTFORMW 砂から長い管状の身体をもたげ、染み出した油に群がる奇怪な魔物が目に留まった
		PRINTFORMW 石油を啜って肥え太る、油ミミズとかいう魔物らしい
		PRINTL
		PRINTFORMW %QST_LEADER_NAME()%らの気配を察したものか、油ミミズたちがこちらに襲い掛かってきた！
	CASE "タールゴーレム"
		PRINTFORML %QST_LEADER_NAME()%が油田の周りを巡回していると……
		PRINTFORMW 石油のすえた匂いがひときわ強くなったかと思うと、足元の砂が急に揺れ始めた
		PRINTFORMW 砂を割って、何かが姿を現わそうとしている……！
		PRINTL
		PRINTFORML 砂の中から現れたのは、粘液質の黒いタールを纏った土巨人……
		PRINTFORMW タールゴーレムとでもいうべき存在だ
		PRINTL
		PRINTFORMW 石油から揮発するマナから生じた存在であろうか
		PRINTFORMW 襲ってくるなら、破壊するしかないだろう……
	CASE "オイルウィスプ"
		PRINTFORMW %QST_LEADER_NAME()%が油田の周りを巡回していると……
		PRINTFORMW ぱち、ぱちと中空で燃える謎の炎の球を発見した
		PRINTL
		PRINTFORMW 石油から揮発したマナを食むために集った精霊の一種、オイルウィスプという名前らしい
		PRINTFORMW 施設や資材に引火するととてもマズイので、申し訳ないが消火させてもらうことにしよう……
	CASE "ペトロスコーピオン"
		PRINTFORMW %QST_LEADER_NAME()%が油田の周りを巡回していると……
		PRINTFORMW 砂の中から、突如長い尻尾が振り上げられ、%QST_LEADER_NAME(1)%の足に鋭利な針を突き立てようとしてきた！
		PRINTL
		PRINTFORMW 咄嗟に身を引いた%QST_LEADER_NAME(1)%の前に、砂の中からキチキチと音を立てて無数のサソリのような魔物が姿を現した
		PRINTFORMW 石油を泳ぐ、と言われるペトロスコーピオンという魔物らしい
		PRINTL
		PRINTFORMW 石油にありつけずに気が立っているのか、こちらに一斉に襲い掛かってきた！
ENDSELECT

path: ERB/QUEST/クエストデータ/QUEST_53_城塞都市ヴェロエア/QUEST_53_城塞都市ヴェロエア_EV160～郊外の窪地.ERB


code:
MAP_NAME = "城塞都市ヴェロエア"
MAP_IMAGE = "q53.webp"
PATH_COLOR = "0x3cff00"
PATH_ALPHA = "0x7d"
NODE_BASE_COLOR = "0xff0000"

path: ERB/QUEST/クエストデータ/QUEST_53_城塞都市ヴェロエア/城塞都市ヴェロエア.txt


code:
<DocumentElement>
  <GMAPDATA>
    <id>638669611570461977</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>新市街入口</NODE_NAME>
    <POS_X>560</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>1,16,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669612140549912</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>異人街</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>0,2,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669612926633107</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>正教の教会</NODE_NAME>
    <POS_X>540</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669613184231331</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>旧市街南門</NODE_NAME>
    <POS_X>480</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>1,4,5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669613788712971</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>城塞への入口</NODE_NAME>
    <POS_X>380</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669616100719860</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>礼拝堂前の広場</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>3,6,7,9,8,10,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669616763193493</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>礼拝堂内部</NODE_NAME>
    <POS_X>240</POS_X>
    <POS_Y>280</POS_Y>
    <PATH_LIST>5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669617106715001</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>初等学院</NODE_NAME>
    <POS_X>280</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>5,8,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669617610552840</id>
    <NODE_ID>8</NODE_ID>
    <NODE_NAME>高等学院</NODE_NAME>
    <POS_X>200</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>5,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669618091835428</id>
    <NODE_ID>9</NODE_ID>
    <NODE_NAME>救貧所</NODE_NAME>
    <POS_X>380</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669618407688893</id>
    <NODE_ID>10</NODE_ID>
    <NODE_NAME>スーク</NODE_NAME>
    <POS_X>160</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>5,11,12,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669619331673643</id>
    <NODE_ID>11</NODE_ID>
    <NODE_NAME>隊商宿</NODE_NAME>
    <POS_X>120</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>10,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669621742077373</id>
    <NODE_ID>12</NODE_ID>
    <NODE_NAME>旧市街北門</NODE_NAME>
    <POS_X>100</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>10,13,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669622294561564</id>
    <NODE_ID>13</NODE_ID>
    <NODE_NAME>ドワーフ居留区入口</NODE_NAME>
    <POS_X>40</POS_X>
    <POS_Y>240</POS_Y>
    <PATH_LIST>12,14,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669622366240748</id>
    <NODE_ID>14</NODE_ID>
    <NODE_NAME>ベルク鉄工所・鍛屋</NODE_NAME>
    <POS_X>60</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>13,15,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669622400719818</id>
    <NODE_ID>15</NODE_ID>
    <NODE_NAME>ベルク鉄工所・事務所</NODE_NAME>
    <POS_X>60</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>14,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638669625418162014</id>
    <NODE_ID>16</NODE_ID>
    <NODE_NAME>郊外の窪地</NODE_NAME>
    <POS_X>480</POS_X>
    <POS_Y>160</POS_Y>
    <PATH_LIST>0,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_53_城塞都市ヴェロエア/城塞都市ヴェロエア.xml


code:
;ノード名
#DEFINE QUEST_57_リスノイ_大通り        0
#DEFINE QUEST_57_リスノイ_官庁街        1
#DEFINE QUEST_57_リスノイ_市場          2
#DEFINE QUEST_57_猟師の小屋             3
#DEFINE QUEST_57_森の入口              10
#DEFINE QUEST_57_森中の空地            11
#DEFINE QUEST_57_森_西部               12
#DEFINE QUEST_57_森_北西部             13
#DEFINE QUEST_57_ゴブリンの巣穴         14

;イベント名
#DEFINE QUEST_57_EV_役場へ向かう        0
#DEFINE QUEST_57_EV_森の下見_前半       1
#DEFINE QUEST_57_EV_森の下見_後半       2
#DEFINE QUEST_57_EV_出撃前の準備        3
#DEFINE QUEST_57_EV_戦闘_森の入口       4
#DEFINE QUEST_57_EV_戦闘_森中の空地     5
#DEFINE QUEST_57_EV_仕切り直し          6
#DEFINE QUEST_57_EV_反撃開始_前半       7
#DEFINE QUEST_57_EV_反撃開始_後半       8
#DEFINE QUEST_57_EV_チーフとの決戦      9
#DEFINE QUEST_57_EV_救出作戦           90
#DEFINE QUEST_57_EV_クエストクリア      99

;時間制限
#DEFINE QUEST_57_タイムリミット         1320 ;22時

;クエストフラグ
#DEFINE QUEST_57_QSTFLG_未クリア           0
#DEFINE QUEST_57_QSTFLG_クエストクリア      1
#DEFINE QUEST_57_QSTFLG_クエスト失敗        2

;ショップ在庫
#DIMS QUEST_57_DIC_SHOP_STOCK_雑貨屋, 10
#DIMS QUEST_57_DIC_SHOP_STOCK_食料品店, 10
#DIMS QUEST_57_DIC_SHOP_STOCK_露店, 10